



Dyad Audit Report

Version 1.0

Oxnightswatch

September 29, 2025

Prepared by: Oxnightswatch

Table of Contents

- Table of Contents
- Protocol Summary
- Disclaimer
- Risk Classification
- Audit Details
 - Scope
 - Roles
- Executive Summary
 - Issues found
- Findings
 - High
 - * [H-1] `VaultManagerV2::liquidate` Permits Users to Liquidate Their Own Positions
 - * [H-2] Collateral Double-Counting When Vault Added to Both `VaultManager` and `KerosineManager`
 - * [H-3] Users Cannot Remove Bounded `KerosineVault` After Addition
 - Medium
 - * [M-1] Return Oracle price lack sanity check for improper return values
 - * [M-2] Calling `vault.oracle()` on `KerosineVault` Reverts Due to Missing Function
 - * [M-3] Collateral Ratio Miscalculation When Liquidating Positions with Unadded Vaults
 - Informational
 - * [I-1] Functions `getVaults` and `hasVault` Return Only Regular (Non-Kerosine) Vaults
- Mocks

Protocol Summary

DYAD is the first truly capital efficient decentralized stablecoin. Traditionally, two costs make stablecoins inefficient: surplus collateral and DEX liquidity. DYAD minimizes both of these costs through Kerosene, a token that lowers the individual cost to mint DYAD.

Disclaimer

Oxnightswatch makes all effort to find as many vulnerabilities in the code in the given time period, but holds no responsibilities for the findings provided in this document. A security audit is not an endorsement of the underlying business or product. The audit was time-boxed and the review of the code was solely on the security aspects of the Solidity implementation of the contracts.

Risk Classification

		Impact		
		High	Medium	Low
Likelihood	High	H	H/M	M
	Medium	H/M	M	M/L
	Low	M	M/L	L

Impact

Audit Details

The findings in this document corresponds to the following commit hash:

```
1
2 https://github.com/code-423n4/2024-04-dyad.git
3 344a58bb84b24d6c281be038a5dff1631426e9c0
```

Scope

```
1
2 ./src/core/VaultManagerV2.sol
3 ./src/core/Vault.kerosine.bounded.sol
4 ./src/core/Vault.kerosine.sol
5 ./src/core/Vault.kerosine.unbounded.sol
6 ./src/core/KerosineManager.sol
7 ./script/deploy/Deploy.V2.s.sol
8 ./src/staking/KerosineDenominator.sol
```

Roles

DYAD Multisig: 0xDeD796De6a14E255487191963dEe436c45995813 Description: Ability to: License new Vault Manager, License new Vaults, Change the kerosene denominator contract, Add new vaults to the Kerosene Manager

Executive Summary

- Start Date: September 23, 2025 6:00 PM
- End Date: Septmeber 28, 2025 8:30 PM

Issues found

Severity	Number of Issues
High	3
Medium	3
Low	0
Informantional	1
Total	7

Findings

In order to run the POC for each finding, create a test file in `/test/`, and then copy and paste each POC in this file and run it

```
1
2 forge test --mt {POC name} -vvv
```

```
1 // SPDX-License-Identifier: SEE LICENSE IN LICENSE
2 pragma solidity ^0.8.0;
3
4 import {DNft} from "../src/core/DNft.sol";
5 import {Dyad} from "../src/core/Dyad.sol";
6 import {USDCMock} from "../USDCMock.sol";
7 import {ERC20Mock} from "../ERC20Mock.sol";
8 import {OracleMock} from "../OracleMock.sol";
9 import {VaultManagerV2} from "../src/core/VaultManagerV2.sol";
10 import {KerosineManager} from "../src/core/KerosineManager.sol";
11 import {Licenser} from "../src/core/Licenser.sol";
12 import {Vault} from "../src/core/Vault.sol";
13 import {BoundedKerosineVault} from "../src/core/Vault.kerosine.bounded.sol";
14 import {UnboundedKerosineVault} from "../src/core/Vault.kerosine.unbounded.sol";
15 import {IAggregatorV3} from "../src/interfaces/IAggregatorV3.sol";
16 import {KerosineDenominatorMock} from "../KerosineDenominatorMock.sol";
17 import {KerosineDenominator} from "../src/staking/KerosineDenominator.sol";
18 import {Kerosine} from "../src/staking/Kerosine.sol";
19
20 import "forge-std/Test.sol";
21 import "forge-std/console2.sol";
22
23 contract POC is Test {
24     address userA = makeAddr("userA");
25     address userB = makeAddr("userB");
26     address owner = address(this);
27     address attacker = makeAddr("attacker");
28
29     Dyad dyad;
30     DNft dNft;
31     Kerosine kerosine;
32     USDCMock usdc;
33     ERC20Mock weth;
34
35     OracleMock usdcOracle;
36     OracleMock wethOracle;
37     KerosineDenominatorMock denominator;
38     VaultManagerV2 manager;
39     KerosineManager kerosineManager;
40     Licenser licenser;
41     Vault regularVaultUSDC;
42     Vault regularVaultWETH;
43     BoundedKerosineVault boundedKerosineVault;
44     UnboundedKerosineVault unboundedKerosineVault;
45
46     uint dNftA;
47     uint dNftB;
48
49     function setUp() public {
50         licenser = new Licenser();
51         kerosineManager = new KerosineManager();
52         dyad = new Dyad(licenser);
53         dNft = new DNft();
54         kerosine = new Kerosine();
55         denominator = new KerosineDenominatorMock(kerosine, address(this));
56         manager = new VaultManagerV2(dNft, dyad, licenser);
57         usdc = new USDCMock("USDC", "USDC");
58         weth = new ERC20Mock("WETH", "WETH");
59         usdcOracle = new OracleMock(1 * 1e8);
60         wethOracle = new OracleMock(3_000 * 1e8);
61         regularVaultUSDC = new Vault(
62             manager,
63             usdc,
64             IAggregatorV3(address(usdcOracle))
65         );
```

```

66     regularVaultWETH = new Vault(
67         manager,
68         weth,
69         IAggregatorV3(address(wethOracle))
70     );
71     unboundedKerosineVault = new UnboundedKerosineVault(
72         manager,
73         kerosine,
74         dyad,
75         kerosineManager
76     );
77     unboundedKerosineVault.setDenominator(
78         KerosineDenominator(address(denominator))
79     );
80
81     boundedKerosineVault = new BoundedKerosineVault(
82         manager,
83         kerosine,
84         kerosineManager
85     );
86     boundedKerosineVault.setUnboundedKerosineVault(unboundedKerosineVault);
87
88     manager.setKeroseneManager(kerosineManager);
89
90     licenseAllVaults();
91     dNftA = dNft.mintNft(userA);
92     dNftB = dNft.mintNft(userB);
93 }
94
95 function licenseAllVaults() internal {
96     licenser.add(address(manager));
97
98     kerosineManager.add(address(regularVaultUSDC));
99     kerosineManager.add(address(regularVaultWETH));
100    kerosineManager.add(address(boundedKerosineVault));
101    kerosineManager.add(address(unboundedKerosineVault));
102
103    licenser.add(address(regularVaultUSDC));
104    licenser.add(address(regularVaultWETH));
105    licenser.add(address(unboundedKerosineVault));
106    licenser.add(address(boundedKerosineVault));
107 }
108 }

```

High

[H-1] VaultManagerV2::liquidate Permits Users to Liquidate Their Own Positions

Description `VaultManagerV2::liquidate` does not verify that the liquidator is different from the owner of the position being liquidated. This omission allows a user to liquidate their own position when the collateralization ratio falls below 150%, which may bypass intended liquidation restrictions.

`src/core/VaultManagerV2.sol#L200-L203`

Impact

Severity: High - because an undercollateralized user can profit instead of being penalized and breaks protocol logic. Likelihood: High - any position can become liquidatable

Proof of Concepts

```

1 function test_userCanLiquidateHimself() public {
2     // userA got minted 160 USDC tokens which are equivalent to $160 at the current price of $1
   per USDC

```

```

3      // then price is changed to make this positon liquidatable and then user liquidate himself
4
5      uint256 userInitialUSDCBalance = 160e6;
6      usdc.mint(userA, userInitialUSDCBalance);
7
8      vm.startPrank(userA);
9      manager.add(dNftA, address(regularVaultUSDC));
10     usdc.approve(address(manager), 160e6);
11     manager.deposit(dNftA, address(regularVaultUSDC), 160e6);
12     manager.mintDyad(dNftA, 100e18, userA);
13     vm.stopPrank();
14
15     vm.roll(block.number + 1);
16
17     // now the price dropped to 80 cents and userA become liquidatable
18     // this create a bad debt to the protocol and the user to be liquidate
19     // retrain all his assets
20     usdcOracle.setPrice(8 * 1e7);
21
22     vm.prank(userA);
23     manager.liquidate(dNftA, dNftA);
24     vm.startPrank(userA);
25     manager.withdraw(dNftA, address(regularVaultUSDC), 160e6, userA);
26
27     uint256 userFinalUSDCBalance = usdc.balanceOf(address(userA));
28
29     assertEq(userInitialUSDCBalance, userFinalUSDCBalance);
30 }

```

Recommended mitigation In `VaultManagerV2::liquidate()` add just the following check

```

1 function liquidate(
2     uint id,
3     uint to
4 )
5     external
6     isValidDNft(id)
7     isValidDNft(to)
8     {
9 +     require(id != to, "self-liquidation");
10     ....
11 }

```

[H-2] Collateral Double-Counting When Vault Added to Both VaultManager and KerosineManager

Description

Regular vaults can be added to both `VaultManager` and `KerosineManager` via `add()` and `addKerosine()`. If a position deposits into a vault that exists in both managers, the same collateral is counted twice when computing total USD value. This results in an inflated collateral balance, allowing a user to appear overcollateralized beyond their actual deposited assets.

The PoC demonstrates this: a single deposit of 160 `USDC` is counted as 320 `USDC` in total value after adding the same vault to both managers.

Root Cause: Regular vaults need to be added to `KerosineManager` for *kerosine price calculation*, but this allow user to add any regular vault twice.

`script/deploy/Deploy.V2.s.sol#L64-L65` `script/deploy/Deploy.V2.s.sol#L93-L96`

Impact

Severity: High — The issue directly affects the integrity of collateral accounting, which is critical for financial safety by inflating user collateral and inflating their position, undermining liquidation logic and exposing the system for undercol-

laterized debt. Likelihood: High — As per the Deploy Script and protocol architecture, the protocol need to add regular vaults to both manangers `assetPrice()` in `UnboundedKerosineVault` to fuction.

Proof of Concepts

```

1 function test_ifRegularVaultIsAddedToVaultLicenserAndKerosineManagerThenCollateralIsDuplicated()
2     public
3     {
4         uint256 depositedAmount = 160e6;
5         usdc.mint(userA, depositedAmount);
6
7         vm.startPrank(userA);
8         manager.add(dNftA, address(regularVaultUSDC));
9         manager.addKerosene(dNftA, address(regularVaultUSDC));
10        usdc.approve(address(manager), depositedAmount);
11        manager.deposit(dNftA, address(regularVaultUSDC), depositedAmount);
12        vm.stopPrank();
13
14        // Assert that total USD value is doubled due to vault being counted twice
15        assertEq(manager.getTotalUsdValue(dNftA), 2 * depositedAmount * 1e12);
16    }

```

Recommended mitigation

The current collateral tracking across multiple managers is prone to double-counting and requires a comprehensive refactor. We recommend reviewing the architecture to ensure each vaults collateral is counted exactly once per position. If immediate refactoring is not feasible, consider implementing a check to prevent the same vault from being added to multiple managers for the same position.

[H-3] Users Cannot Remove Bounded KerosineVault After Addition

Description

Once a `boundedKerosineVault` is added to a position, it cannot be removed, permanently binding the vault to that position. While the **maximum** number of bounded and non-bounded vaults a user can add is limited to **five**, there is no mechanism to remove a bounded vault once added, except via liquidation or other extreme measures. This design reduces flexibility in collateral management.

For example, if a user has added five bounded `KerosineVaults` to their position and then attempts to liquidate another position where the Kerosine vaults differ, the user cannot remove any of their bounded vaults nor add the liquidated bounded Kerosine vault to their position to count toward their collateral.

The root cause is that the protocol requires `id2Asset[id] > 0` for a vault to be considered removable. Since bounded KerosineVaults cannot be withdrawn, this value never reaches zero, preventing removal or reassignment of the vault.

```

1 function removeKerosene(
2     uint id,
3     address vault
4 )
5     external
6     isDNftOwner(id)
7     {
8     @> if (Vault(vault).id2asset(id) > 0) revert VaultHasAssets();
9     if (!vaultsKerosene[id].remove(vault)) revert VaultNotAdded();
10    emit Removed(id, vault);
11    }

```

`src/core/VaultManagerV2.sol#L113`

Impact

Severity: High — because users are permanently bound to bounded KerosineVaults, which can lock collateral, limit capital efficiency, and disrupt protocol operations. Likelihood: High — because this behavior occurs any time a bounded KerosineVault is added and cannot be removed, making it reproducible under normal usage.

Proof of Concepts

```

1 function test_ifUserAddBoundedKerosineVaultItCannotBeRemoved() public {
2     uint256 depositAmount = 100e18;
3
4     // Transfer KEROSINE tokens to the user
5     kerosine.transfer(userA, depositAmount);
6
7     // User deposits KEROSINE into a bounded KerosineVault
8     vm.startPrank(userA);
9     manager.addKerosene(dNftA, address(boundedKerosineVault)); // Add the bounded vault
10    kerosine.approve(address(manager), depositAmount); // Approve tokens for deposit
11    manager.deposit(dNftA, address(boundedKerosineVault), depositAmount); // Deposit tokens
12    vm.stopPrank();
13
14    // Attempting to remove the bounded vault should revert
15    vm.expectRevert();
16    vm.prank(userA);
17    manager.removeKerosene(dNftA, address(boundedKerosineVault));
18 }

```

Recommended mitigation

Introduce a mechanism for positions in boundedKerosineVault to allow the position owner to either:

- Burn the deposited KEROSINE — this can be used to influence or increase the effective asset value within the protocol.
- Transfer back to the vault owner or multisig — enabling the protocol owner to reclaim tokens from positions if necessary.

This provides flexibility and mitigates the issue of permanently locked bounded KerosineVaults.

Medium

[M-1] Return Oracle price lack sanity check for improper return values

Description `VaultManagerV2` relies on the price oracle to determine position collateralization. The contract does not perform sanity checks on the oracles return value. If the oracle returns 0, any position becomes immediately liquidatable, regardless of its actual collateral ratio. This can be exploited by a malicious oracle or misconfigured feed to trigger undesired liquidations.

When vault is called to get price from oracle, `Vault` must perform a sanity check and revert if `answer < 0`

`src/core/Vault.sol#L91-L103`

Impact

Severity: High — because the consequence is blanket, protocol-wide financial loss and integrity failure. Likelihood: Low — because Chainlink mainnet feeds are highly unlikely to return 0 under normal operation.

Proof of Concepts

Using the provided test `test_ifPriceOracleReturnZeroOvercollateralizedUserMayBeLiquidated()` — mint USDC to two users, deposit and mint DYAD so both users are overcollateralized, then force the oracle to return 0 (`usdcOracle.setPrice(0)`). Calling `manager.liquidate(dNftA, dNftB)` succeeds and treats every position as undercollateralized, allowing userB to liquidate userA despite userA being properly collateralized. The PoC demonstrates an oracle value of 0 directly breaks collateral checks and enables mass/external liquidations.

Recommended mitigation

```

1 function assetPrice()
2     public
3     view
4     returns (uint) {

```



```
5      (
6      ,
7      int256 answer,
8      ,
9      uint256 updatedAt,
10     ) = oracle.latestRoundData();
11     if (block.timestamp > updatedAt + STALE_DATA_TIMEOUT) revert StaleData();
12 +   require(answer > 0, "Oracle returned zero");
13     return answer.toUint256();
14 }
```

[M-2] Calling vault.oracle() on KerosineVault Reverts Due to Missing Function

Description KerosineVault contracts do not implement the `.oracle()` function. Any attempt to call `vault.oracle()` on these vaults reverts with “unrecognized function selector”. The `assetPrice()` function in `UnboundedKerosineVault` iterates over all associated vaults, calling `.oracle()` on each. Since non-regular (Kerosine) vaults do not implement `.oracle()`, this call fails, causing the function to revert and breaking price retrieval for KEROSINE positions.

`src/core/Vault.kerosine.unbounded.sol#L50-L68`

Impact

Severity: Medium — no direct fund loss, but usability and protocol reliability are affected. Likelihood: Medium — occurs whenever code iterates over vaults and assumes `.oracle()` exists.

Proof of Concepts

```
1 function test_getKerosinePrice() public {
2     // Transfer KEROSINE to the user
3     kerosine.transfer(userA, 100e18);
4
5     // Deposit KEROSINE into UnboundedKerosineVault
6     vm.startPrank(userA);
7     manager.addKerosene(dNftA, address(unboundedKerosineVault));
8     kerosine.approve(address(manager), 100e18);
9     manager.deposit(dNftA, address(unboundedKerosineVault), 100e18);
10    vm.stopPrank();
11
12    // Retrieve asset price
13    unboundedKerosineVault.assetPrice();
14 }
```

Recommended mitigation

The current protocol architecture does not properly separate vault types, leading to fundamental incompatibilities (e.g., KerosineVaults not supporting `.oracle()`). Addressing this issue will require a comprehensive refactor of the vault management system and cannot be resolved with a minor fix. We recommend that the development team reevaluate and redesign the protocol architecture to ensure consistent and safe handling of all vault types.

[M-3] Collateral Ratio Miscalculation When Liquidating Positions with Unadded Vaults

Description When a liquidator interacts with a position that contains vaults the liquidator has not added to their own position, the liquidators collateral ratio (CR) is calculated without considering collateral from these unadded vaults. If the liquidator has already reached the maximum number of vaults allowed, they cannot claim new collateral unless they exit their current position. This can lead to the liquidator being under-collateralized relative to actual holdings, limiting their ability to mint DYAD or perform other protocol operations.

Impact

Severity: Medium-High — because the liquidators collateral ratio is underestimated, blocking valid minting and restricting capital efficiency. While it does not immediately lead to loss of funds, it creates unfair conditions and systemic inefficiencies. Likelihood: Medium — because mismatched vault participation between liquidators and liquidated positions is a realistic and expected scenario in practice.

Proof of Concepts

```

1 function test_ifLiquidatorLiquidateAPositionWithUnaddedVaultTheLiquidatorCrIsComputedLessThanActual()
2     public
3     {
4         // Setup: normalize WETH price = $1 (same as USDC) to simplify calculations
5         wethOracle.setPrice(1 * 1e8);
6
7         // -----
8         // User A setup
9         // -----
10        // User A deposits $160 of USDC and mints 100 DYAD
11        usdc.mint(userA, 160e6);
12        vm.startPrank(userA);
13        manager.add(dNftA, address(regularVaultUSDC));
14        usdc.approve(address(manager), 160e6);
15        manager.deposit(dNftA, address(regularVaultUSDC), 160e6);
16        manager.mintDyad(dNftA, 100e18, userA);
17        vm.stopPrank();
18
19        // -----
20        // User B setup
21        // -----
22        // User B deposits 310 WETH (priced at $1 = $310) and mints 200 DYAD
23        weth.mint(userB, 310e18);
24        vm.startPrank(userB);
25        manager.add(dNftB, address(regularVaultWETH));
26        weth.approve(address(manager), 310e18);
27        manager.deposit(dNftB, address(regularVaultWETH), 310e18);
28        manager.mintDyad(dNftB, 200e18, userB);
29        vm.stopPrank();
30
31        // -----
32        // Liquidation scenario
33        // -----
34        // USDC price drops to $0.9 User As $160 becomes $144 CR = 1.44 liquidatable
35        usdcOracle.setPrice(0.9 * 1e8);
36        console2.log("Collateral Ratio of UserA: ", manager.collatRatio(dNftA));
37
38        // User B liquidates User A by paying 100 DYAD, receiving ~75% of $144 = ~$108.8
39        // Expected: User Bs total USD collateral = $310 (original) + $108.8 = ~$418.8
40        uint256 UsdUserBValueBefore = manager.getTotalUsdValue(dNftB);
41
42        vm.startPrank(userB);
43        manager.liquidate(dNftA, dNftB);
44        vm.stopPrank();
45
46        // -----
47        // Observed issue
48        // -----
49        // Actual: manager.getTotalUsdValue(dNftB) does NOT include the received USDC collateral
50        // Stays at $310 instead of ~$418.8
51        uint256 UsdUserBValueAfter = manager.getTotalUsdValue(dNftB);
52        assertEq(UsdUserBValueAfter, UsdUserBValueBefore);
53
54        console2.log("Collateral Ratio of UserB: ", manager.collatRatio(dNftB));
55
56        // -----
57        // Downstream impact
58        // -----
59        // Because the $108.8 is ignored, CR is underestimated and minting capacity reduced.
60        // Expected: User B can mint an additional ~72 DYAD safely.
61        // Actual: minting reverts, as if collateral never increased.
62        vm.expectRevert();
63        vm.prank(userB);
64        manager.mintDyad(dNftB, 72e18, userB); // <-- this should not revert

```

```
65 }
```

Recommended mitigation

The liquidation flow should ensure that collateral transferred from a liquidated position is properly recognized in the liquidators accounting, even if it originates from a vault not previously added by the liquidator.

Like:

- Automatically register the vault to the liquidator when collateral is received.

Informational

[I-1] Functions `getVaults` and `hasVault` Return Only Regular (Non-Kerosine) Vaults

Description

The functions `VaultManagerV2::getVaults(uint id)` and `VaultManagerV2::hasVault(uint id, address vault)` currently operate only on regular (non-Kerosine) vaults. There is no handling for KerosineVaults, which may cause confusion for users or integrators who expect these functions to reflect all vault types.

`src/core/VaultManagerV2.sol#L290-L307`

Impact

Informational: No direct financial risk, but may lead to misunderstanding of vault composition or incomplete data when interacting with KerosineVaults. Front-end interfaces or other integrations might incorrectly assume these functions include KerosineVaults, resulting in inaccurate position or collateral displays.

Recommended mitigation

Clearly document that these functions return regular (non-Kerosine) vaults only. Optionally, add separate functions or a combined function that can return or check KerosineVaults as well.

Mocks

Please add the following two mocks contracts to the `./test/`

`/2024-04-dyad/test/KerosineDenominatorMock.sol`

```
1 // SPDX-License-Identifier: MIT
2 pragma solidity =0.8.17;
3
4 // import {Parameters} from "../params/Parameters.sol";
5 import {Kerosine} from "../src/staking/Kerosine.sol";
6
7 contract KerosineDenominatorMock {
8     Kerosine public kerosine;
9     address localOwner;
10
11     constructor(Kerosine _kerosine, address _localOwner) {
12         kerosine = _kerosine;
13         localOwner = _localOwner;
14     }
15
16     function denominator() external view returns (uint) {
17         // @dev: We subtract all the Kerosene in the multi-sig.
18         //       We are aware that this is not a great solution. That is
19         //       why we can switch out Denominator contracts.
20         return kerosine.totalSupply() - kerosine.balanceOf(localOwner);
21     }
22 }
```

and

/2024-04-dyad/test/USCDMock.sol

```
1
2 // SPDX-License-Identifier: MIT
3 pragma solidity =0.8.17;
4
5 import {ERC20} from "@solmate/src/tokens/ERC20.sol";
6 import {SafeTransferLib} from "@solmate/src/utils/SafeTransferLib.sol";
7
8 contract USDCMock is ERC20 {
9     using SafeTransferLib for address;
10
11     constructor(
12         string memory name,
13         string memory symbol
14     ) ERC20(name, symbol, 6) {}
15
16     event Deposit(address indexed from, uint256 amount);
17
18     event Withdrawal(address indexed to, uint256 amount);
19
20     function deposit() public payable virtual {
21         _mint(msg.sender, msg.value);
22
23         emit Deposit(msg.sender, msg.value);
24     }
25
26     function withdraw(uint256 amount) public virtual {
27         _burn(msg.sender, amount);
28
29         emit Withdrawal(msg.sender, amount);
30
31         msg.sender.safeTransferETH(amount);
32     }
33
34     function mint(address to, uint256 amount) external virtual {
35         _mint(to, amount);
36     }
37
38     receive() external payable virtual {
39         deposit();
40     }
41 }
```